# Maintainer: Andrea Dalle Vacche

_realname=ExtUtils-MakeMaker
pkgname=perl-${_realname}
pkgver=7.34
pkgrel=1
pkgdesc="ExtUtils::MakeMaker - Create a module Makefile"
arch=('any')
license=('GPL' 'PerlArtistic')
url="http://search.cpan.org/~bingos/ExtUtils-MakeMaker/"
depends=('perl')
groups=('perl-modules')
options=('!emptydirs')
source=(https://www.cpan.org/authors/id/B/BI/BINGOS/${_realname}-${pkgver}.tar.gz
        001-msysize.patch)
sha256sums=('95f1eb44de480d00b28d031b574ec868f7aeeee199eb5abe5666f6bcbbf68480'
            'd73dce9518b9b810cfaaa118fa775bb922035d65b00976dfb5d88e37109415e3')

prepare() {
  cd ${_realname}-${pkgver}
  patch -p1 -i ${srcdir}/001-msysize.patch
}

build() {
  cd ${_realname}-${pkgver}
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd ${_realname}-${pkgver}
  make test
}

package() {
  cd ${_realname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
